
Dynamické programování

Rekurzivní funkce je taková funkce, která při svém běhu volá sama sebe, často i více
než jednou, což v důsledku může vést na exponenciální algoritmus. Dynamické pro-
gramování je technika, kterou jde z pomalého rekurzivního algoritmu vyrobit pěkný
polynomiální (až na výjimečné případy). Ale nepředbíhejme, nejdříve se podíváme
na jednoduchý příklad rekurze:

Fibonacciho čísla

Budeme počítat n-té číslo Fibonacciho posloupnosti. To je posloupnost, jejíž první
dva členy jsou jedničky a každý další člen je součtem dvou předchozích. Začíná takto:

1 1 2 3 5 8 13 21 34 55 89 . . .

Pro nalezení n-tého členu (ten budeme značit Fn) si napíšeme rekurzivní funkci
Fibonacci(n), která bude postupovat přesně podle definice: zeptá se sama sebe
rekurzivně, jaká jsou dvě předchozí čísla, a pak je sečte. Možná více řekne program:

function Fibonacci(n: integer): integer;
begin
if n <= 2 then
Fibonacci := 1

else
Fibonacci := Fibonacci(n-1) + Fibonacci(n-2)

end;

To, jak funkce volá sama sebe, si můžeme snadno nakreslit třeba pro výpočet čísla F5:

F2 F1

Vidíme, že program se rozvětvuje a tvoří strom volání. Všimněme si také, že některé
podstromy jsou shodné. Zřejmě to budou ty části, které reprezentují výpočet stejného
Fibonacciho čísla – v našem příkladě třeba třetího.

Pokusme se odhadnout časovou složitost Tn naší funkce. Pro n = 1 a n = 2 funkce
skončí hned, tedy v konstantním (řekněme jednotkovém) čase. Pro vyšší n zavolá
sama sebe pro dva předchozí členy plus ještě spotřebuje konstantní čas na sčítání:

Tn ≥ Tn−1 + Tn−2 + const, a proto Tn ≥ Fn.

Tedy na spočítání n-tého Fibonacciho čísla spotřebujeme čas alespoň takový, kolik
je ono číslo samo. Ale jak velké takové Fn vlastně je? Můžeme třeba využít toho, že:

Fn = Fn−1 + Fn−2 ≥ 2 · Fn−2,

z čehož plyne:

Fn ≥ 2n/2.

Funkce Fibonacci má tedy exponenciální časovou složitost, což není nic vítaného
(ukázali jsme sice jen dolní odhad, není však těžké přijít na to, že i v nejhorším
případě poběží exponenciálně pomalu). Ovšem jak jsme už řekli, některé výpočty
opakujeme stále dokola. Nenabízí se proto nic snazšího, než si tyto mezivýsledky
uložit a pak je vytáhnout jako pověstného králíka z klobouku s minimem námahy.

Bude nám k tomu stačit jednoduché pole P o n prvcích, na počátku inicializované
nulami. Kdykoliv budeme chtít spočítat některý člen, nejdříve se podíváme do pole,
zda jsme ho již jednou nespočetli. A naopak jakmile hodnotu spočítáme, hned si ji
do pole poznamenáme:

var P: array[1..MaxN] of integer;
function Fibonacci(n: integer): integer;
begin
if P[n] = 0 then
begin
if n <= 2 then
P[n] := 1

else
P[n] := Fibonacci(n-1) + Fibonacci(n-2)

end;
Fibonacci := P[n]

end;

Podívejme se, jak vypadá strom volání nyní:

Na každý člen posloupnosti se tentokrát ptáme maximálně dvakrát – k výpočtu ho
potřebují dva následující členy. To ale znamená, že funkci Fibonacci zavoláme ma-
ximálně 2n-krát, čili jsme touto jednoduchou úpravou zlepšili exponenciální složitost
na lineární.

Zdálo by se, že abychom získali čas, museli jsme obětovat paměť, ale to není tak
úplně pravda. V prvním příkladu sice nepoužíváme žádné pole, ale při volání funkce
si musíme zapamatovat některé údaje, jako je třeba návratová adresa, parametry
funkce a její lokální proměnné, a na to samotné potřebujeme určitě paměť lineární
s hloubkou vnoření, v našem případě tedy lineární s n.

Určitě vás už také napadlo, že n-té Fibonacciho číslo se dá snadno spočítat i bez
rekurze. Stačí prvky našeho pole P plnit od začátku – kdykoliv známe P [1] =
F1, . . . , Fk = P [k], dokážeme snadno spočítat i P [k + 1] = Fk+1:

function Fibonacci(n: integer): integer;
var
P: array[1..MaxN] of integer;
i: integer;

begin
P[1] := 1;
P[2] := 1;
for i := 3 to n do
P[i] := P[i-1] + P[i-2];

Fibonacci := P[n]
end;

Zopakujme si, co jsme postupně udělali: nejprve jsme vymysleli pomalou rekurzivní
funkci, tu jsme zrychlili zapamatováváním si mezivýsledků a nakonec jsme celou
rekurzi „obrátili narubyÿ a mezivýsledky počítali od nejmenšího k největšímu, aniž
bychom se starali o to, jak se na ně původní rekurze ptala.

V případě Fibonacciho čísel je samozřejmě snadné přijít rovnou na nerekurzivní řeše-
ní (a dokonce si všimnout, že si stačí pamatovat jen poslední dvě hodnoty a paměťo-
vou složitost tak zredukovat na konstantní), ale zmíněný obecný postup zrychlování
rekurze nebo rovnou řešení úlohy od nejmenších podproblémů k těm největším –
obvykle se mu říká dynamické programování – funguje i pro řadu složitějších úloh.
Třeba na tuto:

Problém batohu

Je dáno N předmětů o hmotnostech m1, . . . ,mN (celočíselných) a také číslo M (nos-
nost batohu). Úkolem je vybrat některé z předmětů tak, aby součet jejich hmotností
byl co největší, a přitom nepřekročil M . Předvedeme si algoritmus, který tento pro-
blém řeší v čase O(MN).

Náš algoritmus bude používat pomocné pole A[0 . . .M ] a jeho činnost bude rozdělena
do N kroků. Na konci k-tého kroku bude prvek A[i] nenulový právě tehdy, jestliže
z prvních k předmětů lze vybrat předměty, jejichž součet hmotností je přesně i. Před
prvním krokem (po nultém kroku), jsou všechny hodnoty A[i] pro i > 0 nulové a A[0]
má nějakou nenulovou hodnotu, řekněme −1. Všimněme si, jak kroky algoritmu
odpovídají podúlohám, které řešíme: v prvním kroku vyřešíme podúlohu tvořenou
jen prvním předmětem, ve druhém kroku prvními dvěma předměty, pak prvními
třemi předměty, atd.

Popišme si nyní k-tý krok algoritmu. Pole A budeme procházet od konce, tj. od
i = M . Pokud je hodnota A[i] stále nulová, ale hodnota A[i − mk] je nenulová,
změníme hodnotu uloženou v A[i] na k (později si vysvětlíme, proč zrovna na k).

Nyní si rozmyslíme, že po provedení k-tého kroku odpovídají nenulové hodnoty v po-
li A hmotnostem podmnožin z prvních k předmětů (podmnožina je v podstatě jen

výběr nějaké části předmětů). Pokud je hodnota A[i] nenulová, pak buď byla nenu-
lová před k-tým krokem (a v tom případě odpovídá hmotnosti nějaké podmnožiny
prvních k−1 předmětů) anebo se stala nenulovou v k-tém kroku. Potom ale hodnota
A[i−mk] byla před k-tým krokem nenulová, a tedy existuje podmnožina prvních k−1
předmětů, jejíž hmotnost je i −mk. Přidáním k-tého předmětu k této podmnožině
vytvoříme podmnožinu předmětů hmotnosti přesně i.

Naopak, pokud lze vytvořit podmnožinu X hmotnosti i z prvních k předmětů, pak
takovou podmnožinu X lze buď vytvořit jen z prvních k−1 předmětů, a tedy hodnota
A[i] je nenulová již před k-tým krokem, anebo k-tý předmět je obsažen v takové
množině X. Potom ale hodnota A[i−mk] je nenulová před k-tým krokem (hmotnost
podmnožiny X bez k-tého prvku je i−mk) a hodnota A[i] se stane nenulovou v k-tém
kroku.

Po provedení všech N kroků odpovídají nenulové hodnoty A[i] přesně hmotnostem
podmnožin ze všech předmětů, co máme k dispozici. Speciálně největší index i0 tako-
vý, že hodnota A[i0] je nenulová, odpovídá hmotnosti nejtěžší podmnožiny předmětů,
která nepřekročí hmotnost M . Nalézt jednu množinu této hmotnosti také není ob-
tížné: protože v k-tém kroku jsme měnili nulové hodnoty v poli A na hodnotu k, tak
v A[i0] je uloženo číslo jednoho z předmětů nějaké takové množiny, v A[i0 −mA[i0]]
číslo dalšího předmětu, atd. Zdrojový kód tohoto algoritmu lze nalézt níže.

Časová složitost algoritmu je O(NM), neboť se skládá z N kroků, z nichž každý
vyžaduje čas O(M). Paměťová složitost činí O(N + M), což představuje paměť
potřebnou pro uložení pomocného pole A a hmotností daných předmětů.

var N: integer; { počet předmětů }
M: integer; { hmotnostní omezení }
hmotnost: array[1..N] of integer; { hmotnosti daných předmětů }
A: array[0..M] of integer;
i, k: integer;

begin
A[0]:=-1;
for i:=1 to M do A[i]:=0;
for k:=1 to N do
for i:=M downto hmotnost[k] do
if (A[i-hmotnost[k]]<>0) and (A[i]=0) then
A[i]:=k;

i:=M; while A[i]=0 do i:=i-1;
writeln(’Maximální hmotnost: ’,i);
write(’Předměty v množině:’);
while A[i]<>-1 do begin
write(’ ’,A[i]);
i:=i-hmotnost[A[i]];

end;
writeln;

end.

Cvičení a poznámky

• Proč pole A procházíme pozadu a ne popředu?
• Složitost algoritmu vypadá jako polynomiální, což ve skutečnosti není úplně prav-

da. Závisí totiž na hodnotě M , která má na vstupu délku logM . Algoritmům,
jež jsou polynomiální vůči hodnotám na vstupu (a tedy exponenciální vůči dél-
ce vstupu), se říká pseudopolynomiální. Podrobnosti jsou v kuchařce o těžkých
úlohách.

Nejkratší cesty a Floydův-Warshallův algoritmus

Náš další příklad bude z oblasti grafových algoritmů, ale zkusíme si ho nejdříve říci
bez grafů:

Bylo-nebylo-je N měst. Mezi některými dvojicemi měst vedou (obousměrné) silnice,
jejichž (nezáporné) délky jsou dány na vstupu. Předpokládáme, že silnice se jinde
než ve městech nepotkávají (pokud se kříží, tak mimoúrovňově). Úkolem je spočítat
nejkratší vzdálenosti mezi všemi dvojicemi měst, tj. délky nejkratších cest mezi všemi
dvojicemi měst.

Cestou rozumíme posloupnost měst takovou, že každá dvě po sobě následující města
jsou spojené silnicí, a délka cesty je součet délek silnic, které tato města spojují.
(V grafové terminologii tedy máme daný ohodnocený neorientovaný graf a chceme
zjistit délky nejkratších cest mezi všemi dvojicemi jeho vrcholů.)

Půjdeme na to následovně: Vzdálenosti mezi městy jsou na začátku algoritmu ulo-
ženy ve dvourozměrném poli D, tj. D[i][j] je vzdálenost z města i do města j. Pokud
mezi městy i a j nevede žádná silnice, bude D[i][j] = ∞ (v programu bude tato
hodnota rovna nějakému dostatečně velkému číslu). V průběhu výpočtu si budeme
na pozici D[i][j] udržovat délku nejkratší dosud nalezené cesty mezi městy i a j.

Algoritmus se skládá z N fází. Na konci k-té fáze bude v D[i][j] uložena délka
nejkratší cesty mezi městy i a j, která může procházet skrz libovolná z měst 1, . . . , k.
V průběhu k-té fáze tedy stačí vyzkoušet, zda je mezi městy i a j kratší stávající
cesta přes města 1, . . . , k − 1, jejíž délka je uložena v D[i][j], anebo nová cesta přes
město k.

Pokud nejkratší cesta prochází přes město k, můžeme si ji rozdělit na nejkratší cestu
z i do k a nejkratší cestu z k do j. Délka takové cesty je tedy rovna D[i][k] +
D[k][j]. Takže pokud je součet D[i][k] +D[k][j] menší než stávající hodnota D[i][j],
nahradíme hodnotu na pozici D[i][j] tímto součtem, jinak ji ponecháme.

Z popisu algoritmu přímo plyne, že po N -té fázi je na pozici D[i][j] uložena délka
nejkratší cesty z města i do města j. Protože v každé z N fází algoritmu musíme
vyzkoušet všechny dvojice i a j, vyžaduje každá fáze čas O(N2). Celková časová
složitost našeho algoritmu tedy je O(N3). Co se paměti týče, vystačíme si s polem D
a to má velikost O(N2). Program bude vypadat následovně:

var N: integer; { počet měst }
D: array[1..N] of array[1..N] of longint;
{ délky silnic mezi městy,
D[i][i]=0, místo neexistujících je "nekonečno" }

i, j, k: integer;
begin
for k:=1 to N do
for i:=1 to N do
for j:=1 to N do
if D[i][k]+D[k][j] < D[i][j] then
D[i][j]:=D[i][k] + D[k][j];

end.

Popišme si ještě, jak bychom postupovali, kdybychom kromě vzdáleností mezi městy
chtěli nalézt i nejkratší cesty mezi nimi. To lze jednoduše vyřešit například tak, že
si navíc budeme udržovat pomocné pole E[i][j] a do něj při změně hodnoty D[i][j]
uložíme nejvyšší číslo města na cestě z i do j délky D[i][j] (při změně v k-té fázi je
to číslo k). Máme-li pak vypsat nejkratší cestu z i do j, vypíšeme nejprve cestu z i
do E[i][j] a pak cestu z E[i][j] do j. Tyto cesty nalezneme stejným (rekurzivním)
postupem.

Poznámky

• Popis algoritmu vysloveně svádí k „rejpnutíÿ: Jak víme, že spojením dvou cest,
které provádíme, vznikne zase cesta (tj. že se na ní nemohou nějaké vrcholy opa-
kovat)? To samozřejmě nevíme, ale všimněte si, že kdykoliv by to cesta nebyla,
tak si ji nevybereme, protože původní cesta bez vrcholu k bude vždy kratší nebo
alespoň stejně dlouhá . . . tedy alespoň pokud se v naší zemi nevyskytuje cyklus
záporné délky. (Což, pokud bychom chtěli být přesní, musíme přidat do předpo-
kladů našeho algoritmu.)
• Pozor na pořadí cyklů – program vysloveně svádí k tomu, abychom psali cyklus

pro k jako vnitřní . . . jenže pak samozřejmě nebude fungovat.

Cvičení

• Jak by algoritmus fungoval, kdyby silnice byly jednosměrné?
• Na první pohled nejpřirozenější hodnota, kterou bychom mohli použít pro ∞, je
maxint. To ovšem nebude fungovat, protože∞+∞ přeteče. Stačí maxint div 2?
• Hodnoty v poli si sice přepisujeme pod rukama, takže by se nám mohly poplést

hodnoty z předchozí fáze s těmi z fáze současné. Ale zachrání nás to, že čísla,
o která jde, vyjdou v obou fázích stejně. Proč?

Nejdelší společná podposloupnost

Poslední příklad dynamického programování, který si předvedeme, se bude týkat
posloupností. Mějme dvě posloupnosti čísel A a B. Chceme najít jejich nejdelší
společnou podposloupnost, tedy takovou posloupnost, kterou můžeme získat z A i B
odstraněním některých prvků. Například pro posloupnosti

A = 2 3 3 1 2 3 2 2 3 1 1 2

B = 3 2 2 1 3 1 2 2 3 3 1 2 2 3

je jednou z nejdelších společných podposloupností tato posloupnost:

C = 2 3 1 2 2 3 1 2.

Jakým způsobem můžeme takovou podposloupnost najít? Nejdříve nás asi napadne
vygenerovat všechny podposloupnosti a ty pak porovnat. Jakmile si ale spočítáme,
že všech podposloupností posloupnosti o délce n je 2n (každý prvek nezávisle na
ostatních buď použijeme, nebo ne), najdeme raději nějaké rychlejší řešení.

Zkusme využít následující myšlenku: vyřešíme tento problém pouze pro první prvek
posloupnosti A. Pak najdeme řešení pro první dva prvky A, přičemž využijeme
předchozích výsledků. Takto pokračujeme pro první tři, čtyři, . . . až n prvků.

Nejprve si rozmyslíme, co všechno si musíme v každém kroku pamatovat, abychom
z toho dokázali spočíst krok následující. Určitě nám nebude stačit pamatovat si
pouze nejdelší podposloupnost, jenže množina všech společných podposloupností je
už zase moc velká.

Podívejme se tedy detailněji, jak se změní tato množina při přidání dalšího prvku
k A: Všechny podposloupnosti, které v množině byly, tam zůstanou a navíc přibude
několik nových, končících právě přidaným prvkem. Ovšem my si podposloupnosti
pamatujeme proto, abychom je časem rozšířili na nejdelší společnou podposloupnost,
takže pokud známe nějaké dvě stejně dlouhé podposloupnosti P a Q končící nově
přidaným prvkem v A a víme, že P končí v B dříve než Q, stačí si z nich pamatovat
pouze P , jelikož v libovolném rozšíření Q-čka můžeme Q vyměnit za P a získat tím
stejně dlouhou společnou podposloupnost.

Proto si stačí pro již zpracovaných a prvků posloupnosti A pamatovat pro každou
délku l tu ze společných podposloupností A[1 . . . a] a B délky l, která v B končí
na nejlevějším možném místě, a dokonce nám bude stačit si místo celé podposloup-
nosti uložit jen pozici jejího konce v B. K tomu použijeme dvojrozměrné pole D[a, l].

Ještě si dovolíme jedno malé pozorování: Koncové pozice uložené v poli D se zvětšují
s rostoucí délkou podposloupnosti, čili D[a, l] < D[a, l + 1], protože posloupnosti
délky l + 1 nejsou ničím jiným než rozšířeními posloupností délky l o 1 prvek.

Teď již výpočet samotný: Pokud už známe celý a-tý řádek pole D, můžeme z něj
získat (a + 1)-ní řádek. Projdeme postupně posloupnost B. Když najdeme v B pr-
vek A[a+ 1] (ten právě přidávaný do A), můžeme rozšířit všechny podposloupnosti
končící před aktuální pozicí v B. Nás bude zajímat pouze ta nejdelší z nich, proto-
že rozšířením všech kratších získáme posloupnost, jejíž koncová pozice je větší než
koncová pozice některé posloupnosti, kterou již známe. Rozšíříme tedy tu nejdelší
podposloupnost a uložíme ji místo původní podposloupnosti. Toto provedeme pro
každý výskyt nového prvku v posloupnosti B. Všimněte si, že nemusíme prochá-
zet pole s podposloupnostmi stále od začátku, ale můžeme se v něm posouvat od
nejmenší délky k největší.

Ukážeme si, jak vypadá zaplněné pole hodnotami při řešení problému s posloupnost-
mi z našeho příkladu. Řádky jsou pozice v A, sloupce délky podposloupností.

D 1 2 3 4 5 6 7 8 9 10 11 12
1 2 − − − − − − − − − − −
2 1 5 − − − − − − − − − −
3 1 5 9 − − − − − − − − −
4 1 4 6 11 − − − − − − − −
5 1 2 5 7 12 − − − − − − −
6 1 2 3 7 9 14 − − − − − −
7 1 2 3 7 8 12 − − − − − −
8 1 2 3 7 8 12 13 − − − − −
9 1 2 3 5 8 9 13 14 − − − −
10 1 2 3 4 6 9 11 14 − − − −
11 1 2 3 4 6 9 11 14 − − − −
12 1 2 3 4 6 7 11 12 − − − −

Zbývá popsat, jak z těchto dat zvládneme rekonstruovat hledanou nejdelší společnou
podposloupnost (NSP). Ukážeme si to na našem příkladu: jelikož poslední nenulové
číslo na posledním řádku je v 8. sloupci, má hledaná NSP délku 8. D[12, 8] = 12 říká,
že poslední písmeno NSP je na pozici 12 v posloupnosti B. Jeho pozici v posloupnosti
A určuje nejvyšší řádek, ve kterém se tato hodnota také vyskytuje, v našem případě
je to řádek 12. Druhé písmeno tedy budeme určovat z D[11, 7], třetí z D[9, 6], atd.
Jednou z hledaných podposloupností je:

poslupnost: 2 3 1 2 2 3 1 2
indexy v A: 1 2 4 5 7 9 10 12
indexy v B : 2 5 6 7 8 9 11 12

Ještě trochu konkrétněji:

program Podposloupnost;
var
A, B, C: array[0..MaxN - 1] of Integer;
LA, LB, LC: Integer; { Délky posloupností }
D: array[0..MaxN, 1..MaxN] of Integer;
I, J, L, MaxL, T: Integer;

begin
...
if LA > LB then begin { A bude kratší z obou }
C := A;
A := B;
B := C;
T := LA;
LA := LB;
LB := T;

end;

for I := 1 to LA do
D[0, I] := LB;

L := 0;
MaxL := 0;
for I := 1 to LA do begin
for J := 1 to LA do
D[I, J] := D[I - 1, J];

L := 0;
for J := 0 to LB - 1 do
if B[J] = A[I - 1] then
begin
while (L = 0) or (D[I - 1, L] < J) do
L := L + 1;

if D[I, L] >= J then
D[I, L] := J;

end;
if L > MaxL then MaxL := L;

end;

LC := MaxL;
J := LA;
for I := LC downto 1 do
begin
while D[J - 1, I] = D[J, I] do
J := J - 1;

C[I - 1] := A[J - 1];
J := J - 1;

end;
...

end.

Již zbývá jen odhadnout složitost algoritmu. Časově nejnáročnější byl vlastní vý-
počet hodnot v poli, který se skládá ze dvou hlavních cyklů o délce |A| a |B|, což
jsou délky posloupností A a B. Vnořený cyklus while proběhne celkem maximálně
|A|-krát a časovou složitost nám nezhorší. Můžeme tedy říct, že časová složitost je
O(|A| · |B|). Posloupnosti jsme si prohodili tak, aby první byla ta kratší, protože pak
je maximální délka společné podposloupnosti i počet kroků algoritmu roven délce
kratší posloupnosti a tedy i velikost pole s daty je kvadrát této délky. Paměťovou
složitost odhadneme O(N2 +M), kde N je délka kratší posloupnosti a M té delší.

Cvičení

• Proč jsme si z více posloupností zapamatovali zrovna tu, která v B končí nejle-
vějším možným prvkem?

Martin Mareš a Petr Škoda

Úloha 22-1-3: Sazba

Na vstupu dostanete text a číslo N . Vaším úkolem je zarovnat ho do bloku tak,
aby byl co nejhezčí. Protože krása je věc názoru, zadefinujeme si pro naše účely
vhodné objektivní měřítko: pro každou posloupnost mezer délky k (oddělující slova)
vezmeme číslo (k − 1)2 a tato čísla sečteme přes všechny posloupnosti mezer ve
vysázeném textu. No a sazba je nejhezčí, pokud je tento součet nejmenší.

Slova nelze dělit mezi řádky a máte zaručeno, že se v textu neobjeví slovo delší než N .
Na výstup od vás nechceme vypisovat vytvořené zarovnání, ale pouze minimální výše
popsaný součet pro daný text, tedy číslo.

Například pro text „This is the example you are actually considering.ÿ a N = 28
má program vypsat 12, protože optimální zarovnání je

This is the example you
are actually considering.

a ohodnocení 1 + 1 + 1 + 4 + 1 + 4 = 12.

Úloha 23-2-1: Balíčky balíčků

Chcete poslat poštou petici za lehčí úlohy v KSP organizátorům a co nevidíte. Vý-
hodné nabídky balíčků! Můžete poslat jeden o váze N kg, dva o váze N − 1 kg, tři
o váze N − 2 kg, . . . , N o váze 1 kg, kde N závisí na ročním období, denní hodině
a sjízdnosti silnic. To vás nemusí trápit, N dostane váš program na vstupu.

Dále dostanete váhu H petice v celých kilogramech. Vaším úkolem bude vymyslet,
které nabídky balíčků je třeba vybrat, aby se do nich dohromady vešlo H kg petice,
ale zároveň aby jejich kapacita byla co nejblíže tomuto H.

Je třeba zdůraznit, že „3 balíčky, každý o váze N − 2 kgÿ je jedna nabídka, kterou
jako celek buď přijmete, nebo nepřijmete. Chcete-li poslat 3N − 6 kg, je to ideální
volba.

Chcete-li poslat N − 2 kg a N není úplně malé (třeba N = 100), je lepší zvolit
nabídku „1 balíček o váze N kgÿ, přestože dva kilogramy nevyužijete. Stejně dobré
řešení by pak bylo vybrat „N balíčků o váze 1 kgÿ a nám je jedno, které z takových
dvou stejně dobrých řešení vypíšete.

Chcete-li poslat 100 kg a N = 12, můžete vybrat třeba kombinaci

3 · (N − 2) + 3 · (N − 2) + 5 · (N − 4) = 3 · 10 + 3 · 10 + 5 · 8 = 100

